Computer und Software 1

Christof Köhler

3. Maple – Manipulieren von Ausdrücken, Lösen von Gleichungssystemen

Folien: Balint Aradi

Allgemeine Regel

Eine Warnung im voraus:

Maple macht die Umformung grundsätzlich anders als gedacht.

Grundregel:

- Umformungen erfordern einiges an Kreativität (vieles ausprobieren, bis es geht); eventuell ist man dann doch „zu Fuß“ schneller und benutzt Maple nur zur Probe
- Wenn möglich, sollte Maple Hilfestellung gegeben werden.
- Es sollten immer möglichst kleine Ausdrücke umgeformt werden.
- Oft ist es praktisch, große Ausdrücke in kleinere zu zerlegen, die Teile individuell zu bearbeiten, und diese dann wieder zusammenzusetzen.

expand – Ausmultiplizieren

- Ausmultiplizieren eines Polynoms

```
> expand((x-1)^2*(x+2));
```

$$x^3 - 3\,x + 2$$

- Ausmultiplizieren eines trigonometrischen Polynoms

```
> expand(cos(x+y));
```

$$\cos(x)\cos(y) - \sin(x)\sin(y)$$

- Funktioniert auch für Exponentialfunktionen:

```
> expand(exp(x-y));
```

$$\frac{e^{x}}{e^{y}}$$

```
> expand(tan(x-y));
```

$$\frac{\tan(x) - \tan(y)}{1 + \tan(x)\tan(y)}$$

- Braucht eventuell Hilfestellung

```
> expand(ln(y/(1+y^2)));
```

$$\ln\left(\frac{y}{1 + y^2}\right)$$

Extra Annahme über y

```
> assume(y,real); expand(ln(y/(1+y^2)));
```

$$-\ln(1 + y\sim^2) + \ln(y\sim)$$

assume – Zusätzliche Annahmen über eine Variable

- Definition von Annahmen: **assume()**

```
> sin(x*Pi);
```

$$\sin(x\,\pi)$$

```
> assume(x, integer);
> x;
```

$$x\sim$$

Variable mit Annahmen

```
> sin(x*Pi);
```

$$0$$

- Weitere Annahmen: **additionally()**

```
> cos(x*Pi);
```

$$(-1)^{x\sim}$$

```
> additionally(x/2, integer);
> cos(x*Pi);
```

$$1$$

- Abfragen von Annahmen: **getassumptions()**

```
> getassumptions(x);
```

$$\left\{x\sim::integer, \left(\frac{1}{2}\,x\sim\right)::integer\right\}$$

- Löschen von Annahmen:

```
> x := 'x';
```

$$x := x$$

```
> getassumptions(x);
```

$$\{\}$$

- Lokale Annahmen: **assuming()**

```
> sin(x*Pi) assuming(x, integer);
```

$$0$$

```
> sin(x*Pi);
```

$$\sin(x\,\pi)$$

- Annahme als Relation:

```
> sqrt(x^2);
```

$$\sqrt{x^2}$$

```
> sqrt(x^2) assuming(x>0);
```

$$x$$

combine, radnormal – Zusammenfassen zu einem Term

- **combine()**: Gewissermaßen das Inverse von expand()

```
> expand(cos(x+y));
```

$$\cos(x)\cos(y) - \sin(x)\sin(y)$$

```
> combine(%);
```

$$\cos(x+y)$$

```
> expand(exp(x-y));
```

$$\frac{e^x}{e^y}$$

```
> combine(%);
```

$$e^{(x-y)}$$

```
> expand(ln(x/(1+x^2))) assuming(x, real);
```

$$-\ln(1+x^2) + \ln(x)$$

```
> combine(%) assuming(x, real);
```

$$\ln\left(\frac{x}{1+x^2}\right)$$

- Funktioniert jedoch nicht immer:

```
> expand(tan(x-y));
```

$$\frac{-\tan(y) + \tan(x)}{1 + \tan(y)\tan(x)}$$

```
> combine(%);
```

$$\frac{-\tan(y) + \tan(x)}{1 + \tan(y)\tan(x)}$$

- Für Wurzelausdrücke: **radnormal()**

```
> a := (sqrt(2) - sqrt(3))/(sqrt(2)*sqrt(3))
```

$$a := \frac{1}{6}(\sqrt{2} - \sqrt{3})\sqrt{2}\sqrt{3}$$

```
> combine(a);
```

$$\left(\frac{1}{6}\sqrt{2} - \frac{1}{6}\sqrt{3}\right)\sqrt{6}$$

```
> radnormal(a);
```

$$\frac{1}{3}\sqrt{3} - \frac{1}{2}\sqrt{2}$$

simplify – Vereinfachen von Ausdrücken

- Vereinfacht Ausdrücke, beruhend auf Maples Vorstellung von „einfach“

```
> simplify(cos(x)^2+sin(x)^2);
```

$$1$$

```
> simplify((x^2-2*x+1)/(x-1));
```

$$x - 1$$

```
> simplify(cos(x)^2-sin(x)^2);
```

$$2\cos(x)^2 - 1$$

```
> simplify(2*sin(x)*cos(x));
```

$$2\sin(x)\cos(x)$$

(manchmal geht's mit combine besser)

```
> combine(cos(x)^2-sin(x)^2);
  combine(2*sin(x)*cos(x));
```

$$\cos(2\,x)$$

$$\sin(2\,x)$$

- Vereinfachen mit vorgegebenem Ausdruck

```
> f:=cos(x)^4-3+cos(x)^2;
```

$$f := \cos(x)^4 - 3 + \cos(x)^2$$

```
> simplify(f,{cos(x)^2 = (1+cos(2*x))/2});
```

$$-\frac{9}{4} + \frac{1}{4}\cos(2\,x)^2 + \cos(2\,x)$$

- Sortieren nach Potenzen: **sort()**

```
> sort(2-3*x^2-x+x^4);
```

$$x^4 - 3x^2 - x + 2$$

```
> sort(1-cos(x)^2+cos(x));
```

$$-\cos(x)^2 + \cos(x) + 1$$

- Faktorisieren: **factor()**

```
> factor(x^2-6*x+8);
```

$$(x-2)(x-4)$$

- Herausheben: **collect()**

```
> F:=x*ln(y)-5*x + 3*ln(y)+5*exp(y);
```

$$F := x\ln(y) - 5x + 3\ln(y) + 5\,\mathrm{e}^y$$

```
> collect(F,ln(y));
```

$$(x+3)\ln(y) - 5x + 5\,\mathrm{e}^y$$

```
> F:=a*x*sin(y)+b*a*x/cos(y);
```

$$F := a\,x\sin(y) + \frac{b\,a\,x}{\cos(y)}$$

```
> collect(F,[x,a]);
```

$$\left(\sin(y) + \frac{b}{\cos(y)}\right) a\,x$$

Herausheben von mehreren Termen

isolate

- Isolieren eines Termes auf eine Seite einer Gleichung: **isolate()**

```
> eq := 4 * x * sin(x) = 3;
```

$$eq := 4\,x\sin(x) = 3$$

```
> isolate(eq, sin(x));
```

$$\sin(x) = \frac{3}{4\,x}$$

convert

- Universelles Konversionswerkzeug mit vielen Konversionsmöglichkeiten

Ausdruck zum Konvertieren

```
> convert(sin(x), tan);
```

Art der Konversion

$$\frac{2\tan\left(\frac{1}{2}x\right)}{1+\tan\left(\frac{1}{2}x\right)^2}$$

- Partialbruchzerlegung:

```
> ex := (2*x+1)/(x*(x+1));
```

$$ex := \frac{2x+1}{x(x+1)}$$

```
> convert(ex,parfrac,x);
```

$$\frac{1}{x}+\frac{1}{x+1}$$

- Konversion in Exponentialform

```
> convert(sin(x),exp,x);
> convert(sinh(x),exp,x);
```

$$-\frac{1}{2}I\left(e^{(I\,x)}-\frac{1}{e^{(I\,x)}}\right)$$

$$\frac{1}{2}e^{x}-\frac{1}{2\,e^{x}}$$

convert (#2)

- Englische Einheiten in metrische

```
> convert(1.*acre,metric);convert(1.*MPH,metric);
```

$$4046.856422\ m^2$$

$$\frac{1.609344000\ km}{hr}$$

```
> ?convert[metric]
```

Hilfeseite enthält die Liste der möglichen Einheiten

- Konversion in Stückweise definierte Funktion:

```
> F:=abs(x);
```

$$F := |x|$$

```
> F:=convert(F,piecewise,x);
```

$$F := \begin{cases} -x & x < 0 \\ x & 0 \leq x \end{cases}$$

```
> G:=signum(x);
```

$$G := \mathrm{signum}(x)$$

```
> G:=convert(G,piecewise,x);
```

$$G := \begin{cases} -1 & x < 0 \\ 0 & x = 0 \\ 1 & 0 < x \end{cases}$$

Siehe Hilfe-Seite für weiter Konversionsarten:

```
> ?convert
```

Spezielle Funktionen für Polynome und Polynombrüche

- Grad des Polynoms: **degree()**

```
> degree(2*x^3-x^2+9);
```

$$3$$

- Koeffizienten: **coeffs()**

```
> p := 2*x^3-x^2+9;
```

$$p := 2\,x^3 - x^2 + 9$$

```
> coeffs(p);
```

$$9,\ 2,\ -1$$

Vorsicht, keine Null für fehlende Potenzen!
Reihenfolge unbestimmt

- Zähler und Nenner von Brüchen: **numer()** und **denom()**

```
> b := (x^3-2*x^2-1)/(4*x^2-1);
```

$$b := \frac{x^3 - 2\,x^2 - 1}{4\,x^2 - 1}$$

```
> numer(b);
```

$$x^3 - 2\,x^2 - 1$$

```
> denom(b);
```

$$4\,x^2 - 1$$

Automatische Vereinfachung an der Eingabezeile

- Manchmal hebt Maple automatisch gemeinsame Potenzen heraus:

```
> b := (x^3 - x^2 - x) / (2*x^2-1);
```

$$b := \frac{x^3 - x^2 - x}{2\,x^2 - 1}$$

```
> numer(b);
```

$$x\,(x^2 - x - 1)$$

```
> expand(numer(b));
```

$$x^3 - x^2 - x$$

```
> denom(b);
```

$$2\,x^2 - 1$$

Mit **expand()** kann ursprüngliche Form wieder hergestellt werden

Konversion zwischen Ausdrücken und Funktionen

- Aus einem Ausdruck eine Funktion erstellen: **unapply()**

```
> expr := sin(x) / x;
```

$$expr := \frac{\sin(x)}{x}$$

```
> f := unapply(expr, x);
```

$$f := x \rightarrow \frac{\sin(x)}{x}$$

- Aus einer Funktion einen Ausdruck erstellen: **apply()**

```
> apply(f, 3*x^2);
```

äquivalent

$$\frac{\sin(3\,x^2)}{3\,x^2}$$

```
> f(3*x^2);
```

$$\frac{\sin(3\,x^2)}{3\,x^2}$$

Gleichungslöser

- Variablen können Gleichungen (Ausdrücke) enthalten (wurde schon zuvor genutzt):

```
> gl := x^3 = 2*x;
```

$$gl := x^3 = 2\,x$$

- Auf die linke Seite (left hand side) und rechte Seite (right hand side) der Gleichung kann mit **lhs()** und **rhs()** zugegriffen werden:

```
> lhs(gl);
```

$$x^3$$

```
> rhs(gl);
```

$$2\,x$$

- Eine Gleichung wird mittels **solve()** gelöst (nach einer Variable aufgelöst)

```
> solve(gl, x);
```

$$0, \sqrt{2}, -\sqrt{2}$$

- Numerische (Fließkomma-) Lösung kann mittels **fsolve()** berechnet werden:

```
> fsolve(gl, x);
```

$$-1.414213562, 0., 1.414213562$$

- Bei **mehreren Lösungen** liefert Maple eine **Sequenz**:

```
> mysol1 := %[1];
```

$$mysol1 := -1.414213562$$

Gleichungslöser (#2)

- **Lösen = Nach einer Variable auflösen**
  → Maple kann mit Parametern in der Gleichung umgehen

```
> s1:=solve(a*x^2+b*x+c=0,x);
```

$$s1 := -\frac{b - \sqrt{b^2 - 4\,a\,c}}{2\,a}, -\frac{b + \sqrt{b^2 - 4\,a\,c}}{2\,a}$$

- Numerische Lösung kann auf verschieden Wege ermittelt werden:

```
> s2:=solve(x^2-8*x+14=0,x);
```

$$s2 := 4 + \sqrt{2}, 4 - \sqrt{2}$$

```
> evalf(%);
```

$$5.414213562, 2.585786438$$

```
> solve(x^2-8.0*x+14.0=0, x);
```

$$5.414213562, 2.585786438$$

```
> fsolve(x^2-8*x+14=0, x);
```

$$2.585786438, 5.414213562$$

Numerisches Lösen von Gleichungen

- Bei komplizierten Gleichungen kann Maple keine explizite symbolische Lösung liefern:

```
> s7:=solve(cos(x)-x/10=0,x);
```

$$s7 := \mathrm{RootOf}(\_Z - 10\cos(\_Z))$$

Die Lösungen dieser Gleichung sind die Lösungen der ursprünglichen Gleichung

- Numerische Lösung ist mittels **evalf()** zugänglich:

```
> evalf(s7);
```

$$1.427551779$$

- **Es ist nicht garantiert, dass evalf() alle Lösungen zurückgibt!**

```
> plot(cos(x)-x/10,x=-15..15, thickness=2);
```



7 Lösungen!

Bei der numerischen Lösung einer Gleichung sollte (wenn möglich) vorab mit plot() ermittelt werden, wieviele Lösungen (und in welchem Bereich) zu erwarten sind.

Numerisches Lösen von Gleichungen (#2)

- Der abzusuchende Bereich nach *einer* (!) Lösung kann angegeben werden:

```
> fsolve(cos(x)-x/10=0.,x,-9.5..-8.8);
```

$$-8.966016479$$

- Wird ein Bereich angegeben, wo keine Lösung existiert, liefert Maple keine Antwort.

```
> fsolve(cos(x)-x/10=0.,x,-8.5..-8);
```

$$fsolve\left(\cos(x) - \frac{1}{10}\,x = 0.,\, x,\, -8.5\,..\,-8\right)$$

- Statt eines Bereiches kann ein Startwert angegeben werden, bei welchem Maple die Suche nach der Lösung starten soll.

```
> fsolve(cos(x)-x/10=0.,x=1.5);
```

$$1.427551779$$

- Vorsicht, Lösung kann auch ziemlich weit vom Startpunkt entfernt sein!

```
> fsolve(cos(x)-x/10=0.,x=0);
```

$$7.068891237$$

Numerisches Lösen von Gleichungen (#3)

- Bereits gefundene Lösungen können aus der Lösungssuche ausgeschlossen werden:

```
> plot(cos(x)+x/3, x=0..4);
```




```
> s1:=fsolve(cos(x)+x/3=0,x=3);
```

$$s1 := 2.938100394$$

```
> s2:=fsolve(cos(x)+x/3=0,x=.09,avoid={x=s1});
```

$$s2 := 2.663178883$$

Lösen von Gleichungssystemen

- Analog zu Gleichungen kann Maple auch Gleichungssysteme lösen (nach Variablen auflösen):

```
eq1 := J = 2 * K;
```

$$eq1 := J = 2\,K$$

```
eq2 := K+J = 27;
```

$$eq2 := K + J = 27$$

Gleichungen (als Menge)

auszudrückende Variablen (als Menge)

```
solve({eq1, eq2},{J,K});
```

$$\{K = 9, J = 18\}$$

```
fsolve({eq1,eq2},{J,K});
```

$$\{K = 9., J = 18.\}$$

- Nichtlineare Gleichungssysteme können auch gelöst werden:

```
> eq1 := x^2 - y = 5;
  eq2 := x - y^2 = -13;
```

$$eq1 := x^2 - y = 5$$

$$eq2 := x - y^2 = -13$$

eine explizite Lösung

```
> s := solve({eq1, eq2}, {x, y});
```

$$s := \{x = 3, y = 4\}, \left\{ x = \mathrm{RootOf}(\_Z^3 + 3\,\_Z^2 - \_Z - 4, label = \_L1), y = \mathrm{RootOf}(\_Z^3 + 3\,\_Z^2 - \_Z - 4, label = \_L1)^2 - 5 \right\}$$

weitere Lösungen als Nullstellen eines Polynoms

Numerisches Lösen von Gleichungssystemen

```
> s := solve({eq1, eq2}, {x, y});
```

$$s := \{x = 3, y = 4\}, \left\{ x = \mathrm{RootOf}(\_Z^3 + 3\,\_Z^2 - \_Z - 4, \mathit{label} = \_L1), y = \mathrm{RootOf}(\_Z^3 + 3\,\_Z^2 - \_Z - 4, \mathit{label} = \_L1)^2 - 5 \right\}$$

- Nullstellen des Polynomes via numerische Lösung:

```
> evalf(s[2]);
```

$$\{y = -3.756981175, x = 1.114907541\}$$

- Die Funktion **allvalues()** liefert alle Nullstellen des Polynomes

```
> evalf(allvalues(s[2]));
```

$$\{x = 1.114907542 - 1\ 10^{-10}\ I, y = -3.756981173 - 2.229815084\ 10^{-10}\ I\}, \{x = -2.860805854 + 1\ 10^{-10}\ I, y = 3.184210134 - 5.721611708\ 10^{-10}\ I\}, \{y = -3.427228956 - 2.508203376\ 10^{-10}\ I, x = -1.254101688 + 1\ 10^{-10}\ I\}$$

Numerisches Lösen von Gleichungssystemen (#2)

- Für komplizierte Gleichungssysteme finde Maple oft keine symbolische Lösung:

```
> E1:=cos(x)+y-sqrt(z)=-.191748502;
  E2:=x*y*z=3;
  E3:=x+y+2*z=8;
```

$$E1 := \cos(x) + y - \sqrt{z} = -0.191748502$$

$$E2 := x\,y\,z = 3$$

$$E3 := x + y + 2\,z = 8$$

```
> solve({E1,E2,E3},{x,y,z});
```

Maple gibt überhaupt keine Antwort auf den Aufruf von solve()!

- Man kann in entsprechenden Bereichen nach einer (!) numerischen Lösung suchen:

```
> fsolve({E1,E2,E3},{x,y,z},
  {x=-3..3,y=-3..3,z=-3..3});
```

$$\{x = 1.000000000,\ y = 0.9999999998,\ z = 3.000000000\}$$

- Mit **eliminate()** lassen sich eine oder mehrere Variablen aus dem System eliminieren:

```
> eq1 := J = 2 * K;
  eq2 := K + J = 27;
```

$$eq1 := J = 2\,K$$

$$eq2 := K + J = 27$$

```
> eliminate({eq1, eq2}, {J});
```

$$[\{J = 2\,K\},\ \{K - 9\}]$$

Variable, die entfernt werden soll

Restliche Gleichung(en) nach Zurücksubstitution ('=0' wird nicht ausgeschrieben)

Ausdruck für die Variable

Ungleichungen

- Maple kann auch mit Ungleichungen umgehen:

```
> solve(x^2 > 1);
```

$$RealRange(-\infty, Open(-1)), RealRange(Open(1), \infty)$$

- Nebenbedingungen (via **assume**) werden nur bei den Umformungen berücksichtigt, nicht jedoch bei der Suche nach Lösungen.

```
> solve(x^2 = 1) assuming(x > 0);
```

$$1, -1$$

- Solche Bedingungen müssen als zusätzliche Gleichungen/Ungleichungen formuliert werden:

```
> solve({x^2 =1, x > 0}, x);
```

$$\{x = 1\}$$

- Die Lösungen werden immer auf der Menge der komplexen Zahlen gesucht:

```
> solve((x-2)*(x^2+1) = 0, x);
```

$$2, I, -I$$

- Mit dem **RealDomain**-Paket kann die Suche auf die reelle Zahlen beschränkt werden:

```
> with(RealDomain):
Warning, these protected names have be
unprotected: Im, Re, `^`, arccos, arc
arccsc, arccsch, arcsec, arcsech, arc
arctanh, cos, cosh, cot, coth, csc, cs
expand, limit, ln, log, sec, sech, sig
sinh, solve, sqrt, surd, tan, tanh
> solve((x-2)*(x^2+1) = 0, x);
```

$$2$$

Zwei abschließende Tricks

- Mit **assign()** kann der als Lösung enthaltene Ausdruck als Variablendeklaration verwendet werden

```
> solve({x^2 =1, x > 0}, x);
```

$$\{x = 1\}$$

```
> assign(%);
> x;
```

$$1$$

Äquivalent zu
x := 1

- Die erhaltene Lösung kann via **subs()** in beliebige Ausdrücke eingesetzt werden:

```
> l := solve({x^2 =1, x > 0}, x);
```

$$l := \{x = 1\}$$

```
> subs(l, 3*x+2);
```

$$5$$

Vielen Dank für die Aufmerksamkeit!